Instructions for the use of the of the codes of the near-exact distributions for the likelihood ratio test statistic used to test the sphericity structure of a covariance matrix.

The computational implementation was made using the software Mathematica.

The computational modules described incorporate the results in: Marques, F.J. and Coelho, C.A. (2008). Near-exact distributions for the sphericity likelihood ratio test statistic. Journal of Statistical Planning and Inference, 138, 726-741.

The base modules are the modules available in the web-page:

https://sites.google.com/site/nearexactdistributions/

The modules for the density and cumulative distribution functions of a random variable with a Generalized Near-Integer Gamma (GNIG) distribution denoted ahead, respectively, by `GNIGpdf[r_, b_, l_, a_, w_]` and `GNIGcdf[r_, b_, l_, a_, w_]` have to be downloaded from the above web-page, together with an auxiliary module denoted by `Makec[r_, l_, p_]`. With these modules it is possible to compute, at a given value `w`, the probability density or the cumulative distribution function of a random variable with a GNIG distribution with integer shape parameters given by the vector `r`, non-integer shape parameter `b`, rate parameters give by the vector `l` and by the parameter `a`.

In order to implement the near-exact distributions we need two more auxiliary modules, (i) the first, in Figure 1, corresponds to the moment generating function of the sum of independent log Beta random variables with parameters given by the vector `a` and a vector `b` and (ii) the second module, in Figure 2, gives the `h`th moment of the sum of independent log Beta random variables.

Figure 1: Module to compute the moment generating function of a sum of independent log Beta random variables

```
MGFSLB[a_,b_,t_]:=Product[Gamma[a[[j]]-t]/Gamma[a[[j]]]*
                                    Gamma[a[[j]]+b[[j]]]/Gamma[a[[j]]+b[[j]]-t],{j,1,Length[a]}]
```

Figure 2: Module for the evaluation of the h moment of the sum of independent log Beta random variables

```
MomSLB[a_,b_,h_]:=D[MGFSLB[a,b,t],{t, h}]/.t->0
```

The module to compute the near-exact density and cumulative distribution functions for the negative logarithm of the likelihood ratio test statistic $W = -\log \Lambda$ used to test the sphericity structure of a given covariance matrix are denoted, respectively, by `PDFsphericity[p_,n_,x_,dist_:1]` presented in Figure 3 and `CDFsphericity[p_,n_,x_,dist_:1]` in Figure 4. The 4 arguments of the modules are:

`p` - number of variables;

`n=N-1`, where `N` is the sample size;

x - the running value

dist - indicates which near-exact should be computed

(i) value 1 for the single GNIG distribution;

(ii) value 2 for the mixture of 2 GNIG distributions;

(iii) value 3 for the mixture of 3 GNIG distributions;

(iv) in the absence of specification of this argument the module gives it the default value of 1;

Figure 3: Module for the near-exact probability density function of $-\log \Lambda$

```
CDFsphericity[p_,n_,x_,dist_:1]:=Module[{a,b,bstar,cstar,rj,m,r,vec1,vec2,vec3,vec4,vec5,vec6},
 a = Table[(n + 1)/2 - j/2, {j, 2, p}];
 Clear[vec1, vec2, vec3, vec4, vec5, vec6];
 b = Table[(j - 1)/p + (j - 1)/2, {j, 2, p}];
 bstar = Floor[b]; cstar = Rationalize[b - bstar, 0];
 rj = Flatten[{If[Mod[p, 2] == 0, Floor[p/2 + 1/2], Floor[p/4]],Table[Floor[(p - j)/2 + 1], {j, 3, p}]}];
 If[dist == 1, mm1 = SetPrecision[MomSLB[a + bstar, cstar, 1], 50];
               mm2 = SetPrecision[MomSLB[a + bstar, cstar, 2], 50];
               m = mm1/(mm2 - mm1^2); r = mm1^2/(mm2 - mm1^2);
               GNIGcdf[rj, r, a, m, x],
   If[dist == 2,
            {vec1,vec2,vec3,vec4}={vec1,vec2,vec3,vec4}/.
                   NSolve[Table[SetPrecision[MomSLB[a+bstar,cstar,h],500],{h,1,4}]==
                   Table[Moment[MixtureDistribution[{vec4,1-vec4},{GammaDistribution[vec1,1/vec3],
                                GammaDistribution[vec2,1/vec3]}],h],{h,1,4}],
                                   {vec1,vec2,vec3,vec4},Reals][[1]];
             vec4*GNIGcdf[rj,vec1,a,vec3,x]+(1-vec4)*GNIGcdf[rj,vec2,a,vec3,x],
             {vec1,vec2,vec3,vec4,vec5,vec6}=Select[{vec1,vec2,vec3,vec4,vec5,vec6}/.
                  NSolve[Table[SetPrecision[MomSLB[a+bstar,cstar,h],500],{h,1,6}]==
                         Table[Moment[MixtureDistribution[{vec5,vec6,1-vec5-vec6},{GammaDistribution[vec1,1/vec4],
                                       GammaDistribution[vec2, 1/vec4],GammaDistribution[vec3, 1/vec4]}],h],{h,1,6}],
                         {vec1, vec2, vec3, vec4, vec5, vec6}, Reals],AllTrue[# > 0 &]][[1]];
             vec5*GNIGcdf[rj,vec1,a,vec4,x]+vec6*GNIGcdf[rj,vec2,a,vec4,x]+(1-vec5-vec6)*GNIGcdf[rj,vec3,a,vec4,x]]]]
```

Figure 4: Module for the near-exact cumulative distribution of $W = -\log \Lambda$

```
PDFsphericity[p_,n_,x_,dist_:1]:=Module[{a,b,bstar,cstar,rj,m,r,vec1,vec2,vec3,vec4,vec5,vec6},
 a = Table[(n + 1)/2 - j/2, {j, 2, p}];
 Clear[vec1, vec2, vec3, vec4, vec5, vec6];
 b = Table[(j - 1)/p + (j - 1)/2, {j, 2, p}];
 bstar = Floor[b]; cstar = Rationalize[b - bstar, 0];
 rj = Flatten[{If[Mod[p, 2] == 0, Floor[p/2 + 1/2], Floor[p/4]],Table[Floor[(p - j)/2 + 1], {j, 3, p}]}];
 If[dist == 1, mm1 = SetPrecision[MomSLB[a + bstar, cstar, 1], 50];
              mm2 = SetPrecision[MomSLB[a + bstar, cstar, 2], 50];
              m = mm1/(mm2 - mm1^2); r = mm1^2/(mm2 - mm1^2);
              GNIGpdf[rj, r, a, m, x],
   If[dist == 2,
          {vec1,vec2,vec3,vec4}={vec1,vec2,vec3,vec4}/.
                 NSolve[Table[SetPrecision[MomSLB[a+bstar,cstar,h],500],{h,1,4}]==
                 Table[Moment[MixtureDistribution[{vec4,1-vec4},{GammaDistribution[vec1,1/vec3],
                              GammaDistribution[vec2,1/vec3]}],h],{h,1,4}],
                                 {vec1,vec2,vec3,vec4},Reals][[1]];
           vec4*GNIGpdf[rj,vec1,a,vec3,x]+(1-vec4)*GNIGpdf[rj,vec2,a,vec3,x],
           {vec1,vec2,vec3,vec4,vec5,vec6}=Select[{vec1,vec2,vec3,vec4,vec5,vec6}/.
                NSolve[Table[SetPrecision[MomSLB[a+bstar,cstar,h],500],{h,1,6}]==
                       Table[Moment[MixtureDistribution[{vec5,vec6,1-vec5-vec6},{GammaDistribution[vec1,1/vec4],
                                    GammaDistribution[vec2, 1/vec4],GammaDistribution[vec3, 1/vec4]}],h],{h,1,6}],
                       {vec1, vec2, vec3, vec4, vec5, vec6}, Reals],AllTrue[# > 0 &]][[1]];
           vec5*GNIGpdf[rj,vec1,a,vec4,x]+vec6*GNIGpdf[rj,vec2,a,vec4,x]+(1-vec5-vec6)*GNIGpdf[rj,vec3,a,vec4,x]]]]
```

For example, in Table 6 of the paper where these results were developed we have that the exact quantile of Λ for $p = 5$ and $n = 50$ is equal to 0.6109257783234. Using the module `CDFsphericity[p_,n_,x_,dist_:1]` we have

```
SetPrecision[1-CDFsphericity[5, 50, Rationalize[-Log[0.6109257783234], 0],1], 20]
0.049999943656788132705

SetPrecision[1-CDFsphericity[5, 50, Rationalize[-Log[0.6109257783234], 0],2], 20]
0.050000000027694444807
```

and

```
SetPrecision[1-CDFsphericity[5, 50, Rationalize[-Log[0.6109257783234], 0],3], 20]
0.049999999999968053709.
```